
カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

ver4.0.0 よりグループ診療機能（１つの「日レセ」システムで複数医療機関の運用を可能とする機

能）が実装されました。 

この実装により、ユーザーが独自に作成したプログラム（カスタマイズプログラム）について、次の

修正を行う必要があります。 

 

変更内容 

１．変数「HOSPID」を「HOSPNUM」に変更する。 

２．「HOSPNUM」が定義されているテーブルを検索する場合、必ず「HOSPNUM」を検索条件に含める。 

３．データベースアクセス関数を「ORCDBMAIN」に変更する。 

４．サブプログラム（年齢チェック等）の引数に SPA-AREA を追加する。 

５．初期処理に「SPA-HOSPNUM」の編集を追加する。（バッチプログラム） 

６．バッチ処理強制終了サブプログラム新規作成。（バッチプログラム） 

７．サブプログラム「ORCSJOB」の入力引数に「JOB-HOSPNUM」の編集を追加する。（バッチプログラ

ム） 

８．一時ファイルのファイル名称に「HOSPNUM」の編集を追加する。（バッチプログラム） 

９．ログ出力サブプログラム新規作成。 

 

この他カスタマイズ対象のプログラム（日次・月次統計帳票、レセプトカスタマイズ等）によって固

有の変更があります。 

これらについてはそれぞれの処理について変更内容をまとめた資料を別に作成していますので、そち

らを参照ください。 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

１．変数「HOSPID」を「HOSPNUM」に変更する。 

 

データベースの各テーブルに定義されていた「HOSPID」が「HOSPNUM」に変更になりました。 

 HOSPID(ver3.4.0 以前) HOSPNUM(ver4.0.0 以降) 

項目名 医療機関コード 医療機関番号 

テーブル項目属性 varchar(24) numeric(2,0) 

COBOL 変数項目属性 PIC X(24). PIC 9(2). 

 

（プログラム変更点） 

・変数「HOSPID」を定義している場合は、すべて「HOSPNUM」に変更します。 

・WORKING-STRAGE SECTION 等で「HOSPID」を独自に定義している場合は「HOSPNUM」に変更すると

共に項目属性を“PIC 9(2).”に変更します。 

 

変数変更例 

       * 

       *    一時領域 

        01  WRK-AREA. 

-           03  WRK-HOSPID                          PIC X(24). 

+           03  WRK-HOSPNUM                         PIC 9(02). 

            03  WRK-SYMD                            PIC X(08). 

            03  WRK-EDTYMD1                         PIC X(10). 

            03  WRK-EDTYMD3                         PIC X(22). 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

２．「HOSPNUM」が定義されているテーブルを検索する場合、必ず「HOSPNUM」を検索条

件に含める。 

 

「HOSPNUM」が定義されているテーブルの検索には必ず「HOSPNUM」を検索条件に含めます。 

（プログラム変更点） 

・患者情報、患者番号テーブル等「HOSPID」が定義されていたテーブルは検索条件の「HOSPID」を

「HOSPNUM」に変更します。 

・システム管理マスタ、点数マスタ等 ver4.0.0 より新たに「HOSPNUM」が追加されたテーブルは検

索条件に新たに「HOSPNUM」の編集を追加します。 

 

テーブルの変更内容の詳細は日医標準レセプトソフトデータベーステーブル仕様書を参照くださ

い。 

 

システム管理マスタの検索条件変更例 

        2001-SYSKANRI-SELECT-SEC        SECTION. 

       * 

            MOVE    ZERO            TO  FLG-SYSKANRI 

       * 

            INITIALIZE                  SYS-1001-REC 

            MOVE    "1001"          TO  SYS-1001-KANRICD 

            MOVE    "*"             TO  SYS-1001-KBNCD 

            MOVE    SPA-SYSYMD      TO  SYS-1001-STYUKYMD 

                                        SYS-1001-EDYUKYMD 

+           MOVE    SPA-HOSPNUM     TO  SYS-1001-HOSPNUM 

            MOVE    SYS-1001-REC    TO  MCPDATA-REC 

            MOVE    "tbl_syskanri"  TO  MCP-TABLE 

            MOVE    "key10"         TO  MCP-PATHNAME 

            MOVE    "DBSELECT"      TO  MCP-FUNC 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

３．データベースアクセス関数を「ORCDBMAIN」に変更する。 

 

データベースのアクセスロジックで使用していたサブルーチン「ORCMCPSUB」（あるいは「MCPSUB」,

「MONFUNC」）をすべて「ORCDBMAIN」に変更します。 

「ORCDBMAIN」では入力引数である「SPA-AREA」の変数「SPA-HOSPNUM」の妥当性チェックを行います。

現状では 1以上の数値であることをチェックし、エラーがある場合は返却値「MCP-RC」に"51"を返却

します。 

このため、バッチ処理についてもプログラムの初期処理で「SPA-HOSPNUM」の設定が必須となります。 

ただし、"DBOPEN"や"DBCOMMIT"のように個別のスキーマにアクセスしない呼び出しについては

「SPA-HOSPNUM」の設定は不要です。 

「SPA-HOSPNUM」が未設定の場合やデータベースのアクセスに失敗した場合は syslog にエラーメッセ

ージが出力されます。 

 

患者情報の検索条件変更例 

（プログラム変更点） 

・データベースアクセス関数を「ORCDBMAIN」に変更します。 

・患者情報テーブル等の検索項目「HOSPID」を「HOSPNUM」に変更します。 

・変数「MCP-TABLE」にテーブル名、「MCP-PATHNAME」にパス名を指定し、ＤＢ定義体(tbl_ptinf.db

等)で定義されている検索条件を指定します。 

 

 

（１）ORCMCPSUB から変更を行う場合 

   ①WORKING-STORAGE SECTION より ORCMCPSUB で使用する COPY 句を削除します。 

       * 

            COPY    "MCPDATA.INC". 

            COPY    "MCPAREA". 

-           COPY    "CPORCMCP.INC". 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

   ②患者情報の検索部分を次の内容に変更します。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

            MOVE    ZERO            TO  FLG-PTINF 

       * 

            INITIALIZE                  PTINF-REC 

-           MOVE    SPA-HOSPID      TO  PTINF-HOSPID 

+           MOVE    SPA-HOSPNUM     TO  PTINF-HOSPNUM 

            MOVE    SPA-PTID        TO  PTINF-PTID 

            MOVE    PTINF-REC       TO  MCPDATA-REC 

-           MOVE    "PTINF-KEY"     TO  ORC-DBPATH 

+           MOVE    "tbl_ptinf"     TO  MCP-TABLE 

+           MOVE    "key"           TO  MCP-PATHNAME 

            MOVE    "DBSELECT"      TO  MCP-FUNC 

-           CALL    "ORCMCPSUB"         USING 

+           CALL    "ORCDBMAIN"         USING 

                                        MCPAREA 

-                                       ORCMCPAREA 

                                        MCPDATA-REC 

+                                       SPA-AREA 

            IF    ( MCP-RC          =   ZERO ) 

-               MOVE    "PTINF-KEY"     TO  ORC-DBPATH 

+               MOVE    "tbl_ptinf" TO  MCP-TABLE 

+               MOVE    "key"       TO  MCP-PATHNAME 

                MOVE    "DBFETCH"       TO  MCP-FUNC 

-               CALL    "ORCMCPSUB"         USING 

+               CALL    "ORCDBMAIN"         USING 

                                            MCPAREA 

-                                           ORCMCPAREA 

                                            MCPDATA-REC 

+                                           SPA-AREA 

                IF    ( MCP-RC          =   ZERO ) 

                    MOVE    MCPDATA-REC TO  PTINF-REC 

                ELSE 

                    MOVE    1           TO  FLG-PTINF 

                END-IF 

            ELSE 

                    MOVE    1           TO  FLG-PTINF 

            END-IF 

       * 

-           MOVE    "PTINF-KEY"     TO  ORC-DBPATH 

+           MOVE    "tbl_ptinf"     TO  MCP-TABLE 

+           MOVE    "key"           TO  MCP-PATHNAME 

            MOVE    "DBCLOSECURSOR" TO  MCP-FUNC 

-           CALL    "ORCMCPSUB"         USING 

+           CALL    "ORCDBMAIN"         USING 

                                        MCPAREA 

-                                       ORCMCPAREA 

                                        MCPDATA-REC 

+                                       SPA-AREA 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

（２）MCPSUB から変更を行う場合 

   ①WORKING-STORAGE SECTION より MCPSUB で使用する COPY 句を削除します。 

       * 

            COPY    "MCPDATA.INC". 

            COPY    "MCPAREA". 

-           COPY    "ORCA-DBPATH". 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

   ②患者情報の検索部分を次の内容に変更します。 

            MOVE    ZERO            TO  FLG-PTINF 

       * 

            INITIALIZE                  PTINF-REC 

-           MOVE    SPA-HOSPID      TO  PTINF-HOSPID 

+           MOVE    SPA-HOSPNUM     TO  PTINF-HOSPNUM 

            MOVE    SPA-PTID        TO  PTINF-PTID 

            MOVE    PTINF-REC       TO  MCPDATA-REC 

-           MOVE    PATH-TBL-PTINF-KEY  TO  MCP-PATH 

+           MOVE    "tbl_ptinf"     TO  MCP-TABLE 

+           MOVE    "key"           TO  MCP-PATHNAME 

            MOVE    "DBSELECT"      TO  MCP-FUNC 

-           CALL    "MCPSUB"            USING 

+           CALL    "ORCDBMAIN"         USING 

                                        MCPAREA 

                                        MCPDATA-REC 

+                                       SPA-AREA 

            IF    ( MCP-RC          =   ZERO ) 

-               MOVE    PATH-TBL-PTINF-KEY  TO  MCP-PATH 

+               MOVE    "tbl_ptinf" TO  MCP-TABLE 

+               MOVE    "key"       TO  MCP-PATHNAME 

                MOVE    "DBFETCH"       TO  MCP-FUNC 

-               CALL    "MCPSUB"        USING 

-                                       MCPAREA 

-                                       MCPDATA-REC 

+               CALL    "ORCDBMAIN"         USING 

+                                           MCPAREA 

+                                           MCPDATA-REC 

+                                           SPA-AREA 

                IF    ( MCP-RC          =   ZERO ) 

                    MOVE    MCPDATA-REC TO  PTINF-REC 

                ELSE 

                    MOVE    1           TO  FLG-PTINF 

                END-IF 

            ELSE 

                    MOVE    1           TO  FLG-PTINF 

            END-IF 

       * 

-           MOVE    PATH-TBL-PTINF-KEY  TO  MCP-PATH 

+           MOVE    "tbl_ptinf"     TO  MCP-TABLE 

+           MOVE    "key"           TO  MCP-PATHNAME 

            MOVE    "DBCLOSECURSOR" TO  MCP-FUNC 

-           CALL    "MCPSUB"            USING 

+           CALL    "ORCDBMAIN"         USING 

                                        MCPAREA 

                                        MCPDATA-REC 

+                                       SPA-AREA 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

（３）MONFUNC から変更を行う場合 

   ①患者情報の検索部分を次の内容に変更します。 

            MOVE    ZERO            TO  FLG-PTINF 

       * 

            INITIALIZE                  PTINF-REC 

-           MOVE    SPA-HOSPID      TO  PTINF-HOSPID 

+           MOVE    SPA-HOSPNUM     TO  PTINF-HOSPNUM 

            MOVE    SPA-PTID        TO  PTINF-PTID 

            MOVE    PTINF-REC       TO  MCPDATA-REC 

            MOVE    "tbl_ptinf"     TO  MCP-TABLE 

            MOVE    "key"           TO  MCP-PATHNAME 

            MOVE    "DBSELECT"      TO  MCP-FUNC 

-           CALL    "MONFUNC"           USING 

+           CALL    "ORCDBMAIN"         USING 

                                        MCPAREA 

                                        MCPDATA-REC 

+                                       SPA-AREA 

            IF    ( MCP-RC          =   ZERO ) 

-               MOVE    "tbl_ptinf"     TO  MCP-TABLE 

-               MOVE    "key"           TO  MCP-PATHNAME 

+               MOVE    "tbl_ptinf" TO  MCP-TABLE 

+               MOVE    "key"       TO  MCP-PATHNAME 

                MOVE    "DBFETCH"       TO  MCP-FUNC 

-               CALL    "MONFUNC"       USING 

-                                       MCPAREA 

-                                       MCPDATA-REC 

+               CALL    "ORCDBMAIN"         USING 

+                                           MCPAREA 

+                                           MCPDATA-REC 

+                                           SPA-AREA 

                IF    ( MCP-RC          =   ZERO ) 

                    MOVE    MCPDATA-REC TO  PTINF-REC 

                ELSE 

                    MOVE    1           TO  FLG-PTINF 

                END-IF 

            ELSE 

                    MOVE    1           TO  FLG-PTINF 

            END-IF 

       * 

            MOVE    "tbl_ptinf"     TO  MCP-TABLE 

            MOVE    "key"           TO  MCP-PATHNAME 

            MOVE    "DBCLOSECURSOR" TO  MCP-FUNC 

-           CALL    "MONFUNC"           USING 

+           CALL    "ORCDBMAIN"         USING 

                                        MCPAREA 

                                        MCPDATA-REC 

+                                       SPA-AREA 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

４．サブプログラム（年齢チェック等）の引数に SPA-AREA を追加する。 

 

データベースをアクセスするサブプログラムはすべて引数に「SPA-AREA」を含めるようになりました。 

「SPA-AREA」は COPY 句「COMMON-SPA」で定義されています。 

主だったものでは ORCSAGECHK（年齢チェック）、ORCSJOB(ジョブ管理制御)、ORCHCN03(入院請求書兼

領収書)の引数に SPA-AREA が追加されています。 

また、ORCSPTID(患者ＩＤ取得)はプログラムの引数が変更にされています。 

(変更前) 

           CALL "ORCSPTID"     USING 

                               ORCSPTIDAREA 

                               SPA-1009-TBL 

 

(変更後) 

           CALL "ORCSPTID"     USING 

                               ORCSPTIDAREA 

                               SPA-AREA 

 

独自にサブプログラムを作成されている場合は同様に引数に SPA-AREA を追加してください。 

（プログラム変更点） 

・ORCSAGECHK 等のサブプログラムの引数に SPA-AREA を追加します。 

・独自にサブプログラムを作成している場合は、引数に SPA-AREA を追加します。 

 

 

サブプログラム「ORCSAGECHK」の呼び出し部分変更例 

            CALL    "ORCSAGECHK"        USING 

                                        ORCSAGECHKAREA 

+                                       SPA-AREA 

 

 

SPA-AREA の追加が必要なサブプログラムについては参考資料「サブプログラムの呼び出し形式一覧」

を参照ください。 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

５．初期処理に「SPA-HOSPNUM」の編集を追加する。（バッチプログラム） 

 

バッチプログラム（日次・月次統計帳票、レセプトカスタマイズ等）についてはプログラムの初期処

理で入力パラメタより変数「SPA-HOSPNUM」の編集を行います。「SPA-HOSPNUM」は COPY 句「COMMON-SPA」

で定義されています。 

※入力パラメタは各処理に固有の変更があります。詳細は各処理別の変更内容資料を参照ください。 

 

「SPA-HOSPNUM」の編集は必ず初期処理で行うようにしてください。引数に「SPA-AREA」を持つサブ

プログラムを呼び出す際、サブプログラム側は「SPA-HOSPNUM」の値を期待します。 

 

（プログラム変更点） 

・WORKING-STORAGE SECTION に COPY 句「COMMON-SPA」を追加します。（COMMON-SPA が未定義の場合）

・初期処理に「SPA-AREA」の初期化および変数「SPA-HOSPNUM」の編集を追加します。 

 

 

 

日次・月次統計帳票変更例 

① WORKING-STORAGE SECTION に COPY 句「COMMON-SPA」を追加します。 

        WORKING-STORAGE             SECTION. 

       * 

+           COPY    "COMMON-SPA". 

       * 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

②初期処理に「SPA-AREA」の初期化および「SPA-HOSPNUM」の編集追加します。 

       *    初期 処理 

       ***************************************************************** 

        100-INIT-SEC                SECTION. 

       * 

            INITIALIZE                  FLG-AREA 

            INITIALIZE                  STS-AREA 

            INITIALIZE                  CNT-AREA 

            INITIALIZE                  WRK-AREA 

            INITIALIZE                  KEY-AREA 

            INITIALIZE                  TSRYCD-AREA 

            INITIALIZE                  TPTINF-AREA 

            INITIALIZE                  TOTAL-AREA 

+           INITIALIZE                  SPA-AREA 

       * 

            PERFORM 100-DBOPEN-SEC 

       * 

            UNSTRING   COMMAND-PARAM    DELIMITED  BY  "," 

                                        INTO    LNK-PRTKANRI-RENNUM 

                                                LNK-PRTKANRI-TBL-KEY 

                                                LNK-PRTKANRI-TBL-GROUP 

                                                LNK-PRTKANRI-SHORI-RENNUM 

                                                LNK-PRTKANRI-SRYYM 

                                                LNK-PRTKANRI-SKYYMD 

                                                LNK-PRTKANRI-SHELLID 

                                                LNK-PRTKANRI-PRIORITY 

                                                LNK-PRTKANRI-TERMID 

                                                LNK-PRTKANRI-OPID 

                                                LNK-PRTKANRI-PRTNM 

                                                WRK-PARA-JOBID 

                                                WRK-PARA-SHELLID 

-                                               WRK-PARA-HOSPID 

+                                               WRK-PARA-HOSPNUM 

+                                               RECEERR 

                                                WRK-PARA-SRYYM 

                                                WRK-PARA-NYUGAIKBN 

                                                WRK-PARA-SRYKA 

                                                WRK-PARA-SRYCD (1) 

                                                WRK-PARA-SRYCD (2) 

                                                WRK-PARA-SRYCD (3) 

                                                WRK-PARA-SRYCD (4) 

                                                WRK-PARA-SRYCD (5) 

            END-UNSTRING 

+           MOVE    WRK-PARA-HOSPNUM    TO  SPA-HOSPNUM 

       * 

       *    ステップ管理開始処理 

            MOVE    "STS"           TO  SJOBKANRI-MODE 

            INITIALIZE                  JOBKANRI-REC 

ORCA Project               Copyright(C)2007 Japan Medical Association. All rights reserved. 
11 



カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

６．バッチ処理強制終了サブプログラム新規作成。（バッチプログラム） 

 

バッチ処理強制終了サブプログラム「cobabort」を新規に作成しました。 

日次・月次統計、国保請求書等 dbstub 経由で実行しているモジュール内で何らかの不具合が発生し

(データの不整合等)それ以上処理を実施する必要がない場合には、このサブプログラムを使用してく

ださい。 

このサブプログラムは、引数で渡されたメッセージを syslog に出力した直後にそのモジュールの処

理を強制終了します。 

メッセージの終端には LOW-VALUE を設定してください。 

 

「cobabort」使用例 

      * 

           MOVE    SPACE           TO  WRK-ERRMSG 

           STRING  "ORCR00xx 診療会計情報不整合により異常終了" 

                                       DELIMITED   BY  SIZE 

                   LOW-VALUE           DELIMITED   BY  SIZE 

           INTO    WRK-ERRMSG 

           END-STRING 

      * 

           CALL    "cobabort" USING WRK-ERRMSG 

      * 

 

syslog 

Jul  9 15:48:09 main panda/dbstub[9901]: ORCR00xx 診療会計情報不整合により異常終了 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

７．サブプログラム「ORCSJOB」の入力引数に「JOB-HOSPNUM」の編集を追加する。 

（バッチプログラム） 

（プログラム変更点） 

・サブプログラム「ORCSJOB」の入力引数に「JOB-HOSPNUM」の編集を追加します。 

 

サブプログラム「ORCSJOB」の呼び出し部分変更例 

       * 

       *    ステップ管理開始処理 

            MOVE    "STS"           TO  SJOBKANRI-MODE 

            INITIALIZE                  JOBKANRI-REC 

+           MOVE    SPA-HOSPNUM     TO  JOB-HOSPNUM 

            MOVE    WRK-PARA-JOBID  TO  JOB-JOBID 

            MOVE    WRK-PARA-SHELLID 

                                    TO  JOB-SHELLID 

            MOVE    "ORCBG017"      TO  JOB-PGID 

            MOVE    "診療行為別集計表" 

                                    TO  JOB-SHELLMSG 

            CALL    "ORCSJOB"           USING 

                                        ORCSJOBKANRIAREA 

                                        JOBKANRI-REC 

+                                       SPA-AREA 

       * 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

８．一時ファイルのファイル名称に「HOSPNUM」の編集を追加する。（バッチプログラム） 

 

一時ファイルを作成して編集処理を行っている場合、ファイル名称の編集に「HOSPNUM」を追加しま

す。 

また、一時ファイルの名称を"CPCOMMONDAT2.INC"等の日レセコピー句より編集している場合、出力先

ディレクトリは"/var/tmp"から"/tmp"に変更となります。 

（プログラム変更点） 

・一時ファイルのファイル名称の編集に「HOSPNUM」を追加します。 

 

ファイル名称編集部分変更例 

+           MOVE    WRK-PARA-HOSPNUM TO BG0171-HOSPNUM 

            MOVE    "BG01701"       TO  BG0171-FILE-ID 

            MOVE    LNK-PRTKANRI-TERMID 

                                    TO  BG0171-TERMID 

       * 

+           MOVE    WRK-PARA-HOSPNUM TO BG0172-HOSPNUM 

            MOVE    "BG01702"       TO  BG0172-FILE-ID 

            MOVE    LNK-PRTKANRI-TERMID 

                                    TO  BG0172-TERMID 

       * 

カスタマイズプログラムのグループ診療対応（共通） 

―――――――――――――――――――――――――――――――――――――――――――― 

―――――――――――――――――――――――――――――――――――――――――――― 

９．ログ出力サブプログラム新規作成。 

 

ログ出力サブプログラム「coblog」を新規に作成しました。 

このサブプログラムは、引数で渡されたメッセージを syslog に出力します。 

メッセージの終端には LOW-VALUE を設定してください。 

 

「coblog」使用例 

      * 

           MOVE    SPACE           TO  WRK-MSG 

           STRING  "ORCHC03 テスト"    DELIMITED   BY  SIZE 

                   LOW-VALUE           DELIMITED   BY  SIZE 

           INTO    WRK-MSG 

           END-STRING 

      * 

           CALL    "coblog" USING WRK-MSG 

      * 

 

syslog 

Jul 11 17:57:13 localhost panda/aps-orca23[6105]: ORCHC03 テスト 

